12 行列の固有値問題

物理学の分類の仕方の 1 つに、線形物理学と非線形物理学の 2 つに分類する方法がある。線形物理学は力学のなかの弾性論をはじめ、電磁気学や量子力学 (波動力学) など物理学の大きな部分を占める。これらの分野では、対象とする系に固有のモードの固有値、固有運動を求める問題に帰着する。このような、系に励起される波動という見方で固有振動を問題にする課題は、原子核物理学や固体物理学等の分野にも及び、きわめて適用範囲が広い。

12.1 2 重振子の固有振動

具体例として図 12.1 のような 2 重振子の固有振動を考える。

2 つの質点のつり合いの位置からの変位を x_1, x_2 とすると、系の Lagrangian $\mathcal{L}$ は、

$$\left|\frac{x_1}{l}\right| \ll 1, \quad \left|\frac{x_2}{l}\right| \ll 1 \tag{12.1}$$

の条件のもとでは、

$$\mathcal{L} = \frac{m}{2}\left(\dot{x}_1^2 + \dot{x}_2^2\right) - \frac{1}{2}\frac{mg}{l}\left\{2x_1^2 + (x_2 - x_1)^2\right\} \tag{12.2}$$

で与えられる。

したがって Lagrange の運動方程式は、

$$\frac{d}{dt}\left(\frac{\partial \mathcal{L}}{\partial \dot{x}_i}\right) - \frac{\partial \mathcal{L}}{\partial x_i} = 0 \tag{12.3}$$

から、次の連立方程式となる。

$$\begin{aligned} m\ddot{x}_1 &= -\frac{mg}{l}2x_1 + \frac{mg}{l}(x_2 - x_1) \\ m\ddot{x}_2 &= -\frac{mg}{l}(x_2 - x_1) \end{aligned} \tag{12.4}$$

次にこの系の固有振動、すなわち 2 つの質点が同じ周期で振動するモードを求めてみる。このときの角振動数を ω とし、

$$\ddot{x}_1 = -\omega^2 x_1, \quad \ddot{x}_2 = -\omega^2 x_2 \tag{12.5}$$

を代入する事により以下の式を得る。

$$\begin{aligned} (3\omega_0^2 - \omega^2)x_1 - \omega_0^2 x_2 &= 0 \\ -\omega_0^2 x_1 + (\omega_0^2 - \omega^2)x_2 &= 0 \end{aligned} \tag{12.6}$$

ここで $\omega_0^2 = g/l$ である。この連立方程式は行列を用いて以下のように書き表される。

$$\begin{pmatrix} 3\omega_0^2 & -\omega_0^2 \\ -\omega_0^2 & \omega_0^2 \end{pmatrix}\begin{pmatrix} x_1 \\ x_2 \end{pmatrix} = \omega^2 \begin{pmatrix} x_1 \\ x_2 \end{pmatrix} \tag{12.7}$$

さらに $\lambda = \omega^2/\omega_0^2$ とすると次の式になる。

$$\begin{pmatrix} 3 & -1 \\ -1 & 1 \end{pmatrix}\begin{pmatrix} x_1 \\ x_2 \end{pmatrix} = \lambda \begin{pmatrix} x_1 \\ x_2 \end{pmatrix} \tag{12.8}$$



図 12.1: 2 重振子のモデル。



図 12.2: 質点 n 個をバネで連ねた系。

このように固有振動を求める問題は、式 (12.8) を満たす固有値 λ とそれに対応した固有ベクトル x_1, x_2 を求める問題に帰着する。実際に固有値 λ を求めるには、次の固有値方程式を解けばよい。

$$\begin{vmatrix} 3-\lambda & -1 \\ -1 & 1-\lambda \end{vmatrix} = 0 \tag{12.9}$$

練習

上の固有値方程式を解き 2 つの固有値を求めよ。またそれぞれの固有ベクトルを求め、それぞれがどのような振動運動か図示せよ。

練習

図 12.2 のように質量 m の質点を n 個、左端を固定してバネ係数 k のバネで連ねた系の固有振動を求める問題はどのような行列の固有値問題になるか示せ。この行列が対称行列であることを確認せよ。

12.2 固有値方程式

行列の固有値問題とは、行列 A に対して $A\boldsymbol{x} = \lambda\boldsymbol{x}$、すなわち

$$\begin{pmatrix} a_{11} & a_{12} & \dots & a_{1n} \\ a_{21} & a_{22} & \dots & a_{2n} \\ \vdots & \vdots & \ddots & \vdots \\ a_{n1} & a_{n2} & \dots & a_{nn} \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ \vdots \\ x_n \end{pmatrix} = \lambda \begin{pmatrix} x_1 \\ x_2 \\ \vdots \\ x_n \end{pmatrix} \tag{12.10}$$

を満たす固有値 λ とそれに対応した固有ベクトル $\boldsymbol{x}$ を求める問題である。

このとき固有値 λ は次の固有値方程式

$$\begin{vmatrix} a_{11}-\lambda & a_{12} & \dots & a_{1n} \\ a_{21} & a_{22}-\lambda & \dots & a_{2n} \\ \vdots & \vdots & \ddots & \vdots \\ a_{n1} & a_{n2} & \dots & a_{nn}-\lambda \end{vmatrix} = 0 \tag{12.11}$$

の解として与えられる。この方程式は行列 A が n 次正方行列のとき、λ に関する n 次の方程式である。この方程式を直接解き、n 個の解を求める事は一見簡単に見えるが、実際には n が大きくなると効率が悪く実用的では無い。

現在実用になっている固有値の計算法はどれも相当に大げさなものであり、且つ、利用に際して数値計算上の注意を要するものが多い (汎用のものが無い)。これに対して物理学の固有値問題は、これまで見てきた 2 重振子やバネの連結振動のように、

3 重対角実対称行列

である事が多い。ここで 3 重対角実対称行列とは、

$$A = \begin{pmatrix} \alpha_1 & \beta_1 & 0 & 0 & \dots & 0 & 0 \\ \beta_1 & \alpha_2 & \beta_2 & 0 & \dots & 0 & 0 \\ 0 & \beta_2 & \alpha_3 & \beta_3 & \ddots & 0 & 0 \\ 0 & 0 & \beta_3 & \alpha_4 & \ddots & 0 & 0 \\ \vdots & \vdots & \ddots & \ddots & \ddots & \ddots & 0 \\ 0 & 0 & 0 & 0 & \ddots & \alpha_{n-1} & \beta_{n-1} \\ 0 & 0 & 0 & 0 & 0 & \beta_{n-1} & \alpha_n \end{pmatrix} \tag{12.12}$$

と表される、対角要素とそれと隣り合う要素のみが非 0 で且つ対称 ($a_{ij} = a_{ji}$) のものをいう。

このような 3 重対角行列に関しては、スツルムの定理を用いて固有値を求める事ができる。スツルムの定理の説明については数値計算の教科書に譲るが、

> 「3 重対角実対称行列のような特別な (疎な) 行列に対してなら、固有値を求めるプログラムがすでにあるに違いない (だから、スツルムの定理を知らなくても、行列の固有値問題を数値計算で解けるはずである)」

と考えるのは推奨される態度である。

実際、多くの有用な数値計算プログラムが掲載された、

"Numerical Recipes in Fortran 77, Second Edition", W.H. Press, S.A. Teukolsky, W.T. Vetterling, and B.P. Flannery, (Cambridge Univ. Press, 1993)

を見ると、3 重対角実対称行列の固有値を求めるプログラムが掲載・説明されている。このようなプログラムを用いる場合には、

1. プログラムの動作・制限を正しく理解する
2. 入力・出力を正しく理解する

といった事に注意する必要がある。

オリジナルのプログラムは FORTRAN 77 の文法に厳密に書かれており、若干読みにくいところがあるので、拡張版 FORTRAN 77 の機能を用いて書き直したサンプルプログラム (`tqli.f`) を以下に載せる。プログラムは、

`/home/teacher/z6wt01in/SAMPLE/tqli.f`

として置いてある。

プログラムの先頭部分に、機能と入力・出力に関する説明があり、これを正しく理解すればプログラムの詳細に立ち入らずに道具として使える。まずプログラムが n 次 3 重対角実対称行列の固有値・固有ベクトルを求めるものである事から目的のものである事が分かる (3 重対角実対称行列以外には使えない)。このプログラムを使うには、次元 `n` を指定すると共に、3 重対角実対称行列の要素は `d(n)` 及び `e(n)` の配列に入れて渡せば良いことが分かる。さらに単位行列 `z(n,n)` を渡す必要があることも分かる。計算結果の固有値は `d(n)` に、固有ベクトルは `z(n,n)` に入って戻ってくる。注意として、このプログラムを使用するには関数 `pythag(a,b)` が必要である事が記されているが、このプログラム (`pythag.f`) も以下に載せる ("Numerical Recipes in Fortran 77" に掲載されている)。プログラムは、

`/home/teacher/z6wt01in/SAMPLE/pythag.f`

として置いてある。

サンプルプログラム (tqli.f)

```
      subroutine tqli(d,e,n,z)
      implicit none
c**************************************************
c 機能:
c     n 次 3 重対角実対称行列の固有値・固有ベクトル
c     を求める
c 入力:
c     d(1)-d(n) : 対角行列要素
c     e(2)-e(n) : 対角行列要素に隣り合う要素
c                 (e(1) は任意)
c     n         : 行列の次元
c     z(n,n)    : 単位行列要素
```

```
c 出力:
c     d(1)-d(n) : n 個の固有値
c     z(n,n)    : 規格化された固有ベクトル
c               : i 列目が固有値 d(i) に対応
c 注意:
c     1. 関数: pythag(a,b) が必要
c     2. e の内容は保存されない
c**************************************************
c input/output:
      integer n
      real    d(n),e(n),z(n,n)
c local:
      integer i,iter,k,l,m
      real    b,c,dd,f,g,p,r,s,pythag
c begin:
      do i=2,n
         e(i-1)=e(i)
      end do
      e(n)=0.
      do l=1,n
         iter=0
 1       do m=l,n-1
            dd=abs(d(m))+abs(d(m+1))
            if (abs(e(m))+dd.eq.dd) goto 2
         end do
         m=n
 2       if (m.ne.l)then
            if (iter.eq.30) pause 'too many iterations in tqli'
            iter=iter+1
            g=(d(l+1)-d(l))/(2.*e(l))
            r=pythag(g,1.)
            g=d(m)-d(l)+e(l)/(g+sign(r,g))
            s=1.
            c=1.
            p=0.
            do i=m-1,l,-1
               f=s*e(i)
               b=c*e(i)
               r=pythag(f,g)
               e(i+1)=r
               if(r.eq.0.)then
                  d(i+1)=d(i+1)-p
```

```
            e(m)=0.
            goto 1
          endif
          s=f/r
          c=g/r
          g=d(i+1)-p
          r=(d(i)-g)*s+2.*c*b
          p=s*r
          d(i+1)=g+p
          g=c*r-b
          do  k=1,n
            f=z(k,i+1)
            z(k,i+1)=s*z(k,i)+c*f
            z(k,i)=c*z(k,i)-s*f
          end do
        end do
        d(l)=d(l)-p
        e(l)=g
        e(m)=0.
        goto 1
      endif
    end do
    return
    end
```

サンプルプログラム (pythag.f)

```
      real function pythag(a,b)
      implicit none
c input/output:
      real a,b
c local:
      real absa,absb
c begin:
      absa=abs(a)
      absb=abs(b)
      if(absa.gt.absb)then
        pythag=absa*sqrt(1.+(absb/absa)**2)
      else
        if(absb.eq.0.)then
          pythag=0.
        else
```

```
        pythag=absb*sqrt(1.+(absa/absb)**2)
      endif
    endif
    return
    end
```

練習

2 重振子の例題の固有値並びに固有ベクトルを数値計算により求めよ (注:行列は 3 重対角実対称行列である)。

課題

図 12.2 のようなバネの連結振動の場合の、固有値方程式を数値計算により解き、固有値並びに固有ベクトルを求めよ。それぞれどのような振動運動か図示せよ。

計算・メモ用余白